Finding the minimal polynomial for each algebraic element over $\mathbb{Q}$. Can someone check whether the following are the correct minimal polynomials for each root?
For root $\sqrt{3}+\sqrt[3]{5}$, I got $p(x)=x^{6}-9x^{4}-10x^{3}+27x^{2}-90x - 27$.
For root $\cos\theta +i\sin\theta$, where $\theta =\frac{2\pi}{n}$ for $n\geq 1$, I got $p(x)=x^{2}-2x\cos\theta +1$.
For root $\sqrt{\sqrt[3]{2}-i}$, I got $p(x)=x^{12}-15x^{8}-4x^{6}+3x^{4}+12x^{2}+5$.
I tried using a calculator to see whether the equations become zero if I substitute the roots, but no calculator I can find supports such lengthy equations.
 A: Just use sage. Here is the code, just checking:
First polynomial is ok:
sage: (3^(1/2) + 5^(1/3) ).minpoly()
x^6 - 9*x^4 - 10*x^3 + 27*x^2 - 90*x - 2

In fact, sage can work explicitly in the tower of fields:
sage: K.<a> = QuadraticField(3)
sage: R.<X> = PolynomialRing(K)
sage: L.<b> = K.extension(X^2+3)
sage: S.<Y> = PolynomialRing(L)
sage: M.<c> = L.extension(x^3-5)
sage: a^2, b^2, c^3
(3, -3, 5)
sage: third_roots_of_one = [ 1, (-1+b)/2, (-1-b)/2 ]
sage: secnd_roots_of_one = [ 1, -1 ]
sage: T.<Z> = PolynomialRing(M)
sage: prod( [ Z - (r2*a + r3*c) 
....:         for r2 in secnd_roots_of_one
....:         for r3 in third_roots_of_one ] )

Z^6 - 9*Z^4 - 10*Z^3 + 27*Z^2 - 90*Z - 2

For the second question, if we need the minimal polynomial over $\Bbb Q$, the result should be a cyclotomic polynomial. For instance, sage again:
sage: N = 9
sage: exp( 2*pi*i/N ).minpoly()
x^6 + x^3 + 1
sage: cyclotomic_polynomial(N)
x^6 + x^3 + 1

The third one:
sage: sqrt( 2^(1/3) - i ).minpoly()
x^12 + 3*x^8 - 4*x^6 + 3*x^4 + 12*x^2 + 5

Again, one can manually / humanly build the tower of fields, by starting with $\Bbb Q$, and adjoining $i$, then $2^{1/3}$ and its conjugates, then the square root we need.
